One-Degree-of-Freedom Tests

- Test for group $\times$ occasion interactions has
  \[(\text{number of groups} - 1) \times \text{number of occasions} - 1\]
degrees of freedom.

- This can dilute the significance of a departure from the null hypothesis.

- We can focus the test on departures of a particular form.
For example:

- if we expect all responses after the baseline to respond similarly:

  \[ \mu_{\text{after baseline}} = \mu_{\text{baseline}} + \text{shift}, \]

  where \textit{shift} depends on treatment group, we estimate the difference of \((\mu_1 + \mu_4 + \mu_6)/3 - \mu_0\) between treated and placebo groups:

  \[
  \left[ (\mu_{S,1} + \mu_{S,4} + \mu_{S,6})/3 - \mu_{S,0} \right] - \left[ (\mu_{P,1} + \mu_{P,4} + \mu_{P,6})/3 - \mu_{P,0} \right]
  \]

- or we can focus on the area under the curve

  \[
  2 \times (\mu_1 - \mu_0) + 2.5 \times (\mu_4 - \mu_0) + 1 \times (\mu_6 - \mu_0)
  \]

  and again estimate the difference between the groups.
Test these hypotheses in PROC MIXED by adding the statements

contrast 'response vs baseline'
  treatment * week  -1 -1 -1  3
  1  1  1 -3 /
  chisq;
contrast 'AUC'
  treatment * week  -2 -2.5 -1  5.5
  2  2.5  1 -5.5 /
  chisq;

The additional output is

<table>
<thead>
<tr>
<th>Label</th>
<th>Num</th>
<th>Den</th>
<th>Chi-Square</th>
<th>F Value</th>
<th>Pr &gt; ChiSq</th>
<th>Pr &gt; F</th>
</tr>
</thead>
<tbody>
<tr>
<td>response vs baseline</td>
<td>1</td>
<td>98</td>
<td>67.32</td>
<td>67.32</td>
<td>&lt;.0001</td>
<td>&lt;.0001</td>
</tr>
<tr>
<td>AUC</td>
<td>1</td>
<td>98</td>
<td>80.52</td>
<td>80.52</td>
<td>&lt;.0001</td>
<td>&lt;.0001</td>
</tr>
</tbody>
</table>
Exploiting the Baseline

- Profile analysis tests for any differences in change over time.

- With baseline data and randomized groups, we know that the baseline responses can differ between groups only by sampling variation.

- We can exploit this knowledge by
  - converting the later responses into differences from baseline, or
  - treating the baseline measurement as a covariate instead of a response.
Convert to differences:

...  
data tlc_uni;
/* Convert the data from multivariate form (one record per subject, with  
 * 4 responses), to univariate form (one record per response, with a new  
 * variable 'week' to identify the occasion; differences from baseline */  
    set tlc;
    drop y0 -- y6;
    week = 1; y = y1 - y0; output;
    week = 4; y = y4 - y0; output;
    week = 6; y = y6 - y0; output;
run;

proc sort data = tlc_uni;
    by treatment;

proc mixed data = tlc_uni order = data;
    class child treatment week;
    model y = treatment week treatment * week /
            solution chisq;
    repeated week / subject = child type = un;
run;
The key output is

Mixed Model Analysis of TLC data
Differences from baseline

Type 3 Tests of Fixed Effects

<table>
<thead>
<tr>
<th>Effect</th>
<th>Num DF</th>
<th>Den DF</th>
<th>Chi-Square</th>
<th>F Value</th>
<th>Pr &gt; ChiSq</th>
<th>Pr &gt; F</th>
</tr>
</thead>
<tbody>
<tr>
<td>treatment</td>
<td>1</td>
<td>98</td>
<td>67.32</td>
<td>67.32</td>
<td>&lt;.0001</td>
<td>&lt;.0001</td>
</tr>
<tr>
<td>week</td>
<td>2</td>
<td>98</td>
<td>22.20</td>
<td>11.10</td>
<td>&lt;.0001</td>
<td>&lt;.0001</td>
</tr>
<tr>
<td>treatment*week</td>
<td>2</td>
<td>98</td>
<td>39.76</td>
<td>19.88</td>
<td>&lt;.0001</td>
<td>&lt;.0001</td>
</tr>
</tbody>
</table>

Note that the main effect of treatment is now of substantive interest. In fact, it is the same as the simpler one-degree-of-freedom test described above. These treatment and treatment × week Chi-Squares sum to the interaction term in the previous analysis.
Covariate approach:

... data tlc_uni;
/* Convert the data from multivariate form (one record per subject, with
 * 4 responses), to univariate form (one record per response, with a new
 * variable ’week’ to identify the occasion; keep y0 as a covariate */
 set tlc;
 drop y1 -- y6;
 week = 1; y = y1; output;
 week = 4; y = y4; output;
 week = 6; y = y6; output;
run;

proc sort data = tlc_uni;
 by treatment;

proc mixed data = tlc_uni order = data;
 class child treatment week;
 model y = treatment week treatment * week y0 /
 solution chisq;
 repeated week / subject = child type = un;
run;
The key output is

Mixed Model Analysis of TLC data
Use baseline data as a covariate

The Mixed Procedure

Solution for Fixed Effects

| Effect   | treatment | week | Estimate | Error | DF | t Value | Pr > |t| |
|----------|-----------|------|----------|-------|----|---------|------|---|
| y0       |           |      | 0.8045   | 0.09390 | 97 | 8.57    | <.0001 |

Type 3 Tests of Fixed Effects

<table>
<thead>
<tr>
<th>Effect</th>
<th>Num DF</th>
<th>Den DF</th>
<th>Chi-Square</th>
<th>F Value</th>
<th>Pr &gt; ChiSq</th>
<th>Pr &gt; F</th>
</tr>
</thead>
<tbody>
<tr>
<td>treatment</td>
<td>1</td>
<td>97</td>
<td>68.58</td>
<td>68.58</td>
<td>&lt;.0001</td>
<td>&lt;.0001</td>
</tr>
<tr>
<td>week</td>
<td>2</td>
<td>97</td>
<td>22.20</td>
<td>11.10</td>
<td>&lt;.0001</td>
<td>&lt;.0001</td>
</tr>
<tr>
<td>treatment*week</td>
<td>2</td>
<td>97</td>
<td>39.76</td>
<td>19.88</td>
<td>&lt;.0001</td>
<td>&lt;.0001</td>
</tr>
<tr>
<td>y0</td>
<td>1</td>
<td>97</td>
<td>73.41</td>
<td>73.41</td>
<td>&lt;.0001</td>
<td>&lt;.0001</td>
</tr>
</tbody>
</table>
The week and treatment $\times$ week tests are the same as in the “deviation from baseline” analysis, but the main effect of treatment is slightly more significant.

The $y_0$ coefficient of 0.8045 is more than 2 standard deviations less than 1 (significant at the 5% level).
Constraining baseline means to be equal:

... 

data tlc_uni;
   set tlc;
   drop y0 -- y6;
   week = 1; y = y1; output;
   week = 4; y = y4; output;
   week = 6; y = y6; output;
   /* put baseline in its own group: */
   week = 0; y = y0; treatment = 'Z'; output;
run;

proc sort data = tlc_uni;
   by treatment;

proc mixed data = tlc_uni order = data;
   class child treatment week;
   model y = week treatment * week /* no main effect of treatment */
      / solution chisq;
   repeated week / subject = child type = un;
run;
The key output is

Mixed Model Analysis of TLC data, baseline means equal

The Mixed Procedure

Solution for Fixed Effects

| Effect          | treatment | week | Estimate | Error  | DF | t Value | Pr > |t| |
|-----------------|-----------|------|----------|--------|----|---------|------|---|
| Intercept       |           |      | 26.4060  | 0.4999 | 99 | 52.83   | <.0001|
| week 1          | 1         |      | -1.6445  | 0.7823 | 99 | -2.10   | 0.0381|
| week 4          | 4         |      | -2.2313  | 0.8073 | 99 | -2.76   | 0.0068|
| week 6          | 6         |      | -2.6420  | 0.8864 | 99 | -2.98   | 0.0036|
| treatment*week A 1 | 1   |      | -11.3410 | 1.0930 | 99 | -10.38  | <.0001|
| treatment*week A 4 | 4   |      | -8.7653  | 1.1312 | 99 | -7.75   | <.0001|
| treatment*week A 6 | 6   |      | -3.1199  | 1.2507 | 99 | -2.49   | 0.0143|
| treatment*week P 1 | 1   |      | 0        |        | .  |         |      |
| treatment*week P 4 | 4   |      | 0        |        | .  |         |      |
| treatment*week P 6 | 6   |      | 0        |        | .  |         |      |
| treatment*week Z 0 | 0   |      | 0        |        | .  |         |      |
### Type 3 Tests of Fixed Effects

<table>
<thead>
<tr>
<th>Effect</th>
<th>Num DF</th>
<th>Den DF</th>
<th>Chi-Square</th>
<th>F Value</th>
<th>Pr &gt; ChiSq</th>
<th>Pr &gt; F</th>
</tr>
</thead>
<tbody>
<tr>
<td>week</td>
<td>3</td>
<td>99</td>
<td>184.61</td>
<td>61.54</td>
<td>&lt;.0001</td>
<td>&lt;.0001</td>
</tr>
<tr>
<td>treatment*week</td>
<td>3</td>
<td>99</td>
<td>111.96</td>
<td>37.32</td>
<td>&lt;.0001</td>
<td>&lt;.0001</td>
</tr>
</tbody>
</table>

This approach:

- exploits the information about baseline data;
- is as powerful as any other;
- uses data for a subject even when the baseline is missing.